// File: POLYPRIDE.sol



/*
*  _____   ____  _  __     ____  __  ____  _____  _____  _    _ _____ _____ 
* |  __ \ / __ \| | \ \   / /  \/  |/ __ \|  __ \|  __ \| |  | |_   _/ ____|
* | |__) | |  | | |  \ \_/ /| \  / | |  | | |__) | |__) | |__| | | || |     
* |  ___/| |  | | |   \   / | |\/| | |  | |  _  /|  ___/|  __  | | || |     
* | |    | |__| | |____| |  | |  | | |__| | | \ \| |    | |  | |_| || |____ 
* |_|___  \____/|______|_|__|_|__|_|\____/|_| _\_\_|    |_|  |_|_____\_____|
* |  __ \|  __ \|_   _|  __ \|  ____| |  ____| |        /\   / ____|/ _____| 
* | |__) | |__) | | | | |  | | |__    | |__  | |       /  \ | |  __| (___   
* |  ___/|  _  /  | | | |  | |  __|   |  __| | |      / /\ \| | |_ |\___ \  
* | |    | | \ \ _| |_| |__| | |____  | |    | |____ / ____ \ |__| |____) | 
* |_|    |_|  \_\_____|_____/|______| |_|    |______/_/    \_\_____|_____/ 
*
* 1969 fully on-chain pride flags that can be morphed between one of 12 designs
*                project by smol farm | contract by Thorne
*/

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol


// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol


// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol


// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol


// OpenZeppelin Contracts v4.4.1 (security/Pausable.sol)

pragma solidity ^0.8.0;


/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        require(!paused(), "Pausable: paused");
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        require(paused(), "Pausable: not paused");
        _;
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol


// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;


/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Address.sol


// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol


// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol


// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol


// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;


/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol


// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;


/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol


// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;


/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: contracts/ERC721A.sol


// Creator: Chiru Labs

pragma solidity ^0.8.4;








error ApprovalCallerNotOwnerNorApproved();
error ApprovalQueryForNonexistentToken();
error ApproveToCaller();
error ApprovalToCurrentOwner();
error BalanceQueryForZeroAddress();
error MintToZeroAddress();
error MintZeroQuantity();
error OwnerQueryForNonexistentToken();
error TransferCallerNotOwnerNorApproved();
error TransferFromIncorrectOwner();
error TransferToNonERC721ReceiverImplementer();
error TransferToZeroAddress();
error URIQueryForNonexistentToken();

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension. Built to optimize for lower gas during batch mints.
 *
 * Assumes serials are sequentially minted starting at _startTokenId() (defaults to 0, e.g. 0, 1, 2, 3..).
 *
 * Assumes that an owner cannot have more than 2**64 - 1 (max value of uint64) of supply.
 *
 * Assumes that the maximum token id cannot exceed 2**256 - 1 (max value of uint256).
 */
contract ERC721A is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Compiler will pack this into a single 256bit word.
    struct TokenOwnership {
        // The address of the owner.
        address addr;
        // Keeps track of the start time of ownership with minimal overhead for tokenomics.
        uint64 startTimestamp;
        // Whether the token has been burned.
        bool burned;
    }

    // Compiler will pack this into a single 256bit word.
    struct AddressData {
        // Realistically, 2**64-1 is more than enough.
        uint64 balance;
        // Keeps track of mint count with minimal overhead for tokenomics.
        uint64 numberMinted;
        // Keeps track of burn count with minimal overhead for tokenomics.
        uint64 numberBurned;
        // For miscellaneous variable(s) pertaining to the address
        // (e.g. number of whitelist mint slots used).
        // If there are multiple variables, please pack them into a uint64.
        uint64 aux;
    }

    // The tokenId of the next token to be minted.
    uint256 internal _currentIndex;

    // The number of tokens burned.
    uint256 internal _burnCounter;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to ownership details
    // An empty struct value does not necessarily mean the token is unowned. See _ownershipOf implementation for details.
    mapping(uint256 => TokenOwnership) internal _ownerships;

    // Mapping owner address to address data
    mapping(address => AddressData) private _addressData;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
        _currentIndex = _startTokenId();
    }

    /**
     * To change the starting tokenId, please override this function.
     */
    function _startTokenId() internal view virtual returns (uint256) {
        return 0;
    }

    /**
     * @dev Burned tokens are calculated here, use _totalMinted() if you want to count just minted tokens.
     */
    function totalSupply() public view returns (uint256) {
        // Counter underflow is impossible as _burnCounter cannot be incremented
        // more than _currentIndex - _startTokenId() times
        unchecked {
            return _currentIndex - _burnCounter - _startTokenId();
        }
    }

    /**
     * Returns the total amount of tokens minted in the contract.
     */
    function _totalMinted() internal view returns (uint256) {
        // Counter underflow is impossible as _currentIndex does not decrement,
        // and it is initialized to _startTokenId()
        unchecked {
            return _currentIndex - _startTokenId();
        }
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view override returns (uint256) {
        if (owner == address(0)) revert BalanceQueryForZeroAddress();
        return uint256(_addressData[owner].balance);
    }

    /**
     * Returns the number of tokens minted by `owner`.
     */
    function _numberMinted(address owner) internal view returns (uint256) {
        return uint256(_addressData[owner].numberMinted);
    }

    /**
     * Returns the number of tokens burned by or on behalf of `owner`.
     */
    function _numberBurned(address owner) internal view returns (uint256) {
        return uint256(_addressData[owner].numberBurned);
    }

    /**
     * Returns the auxillary data for `owner`. (e.g. number of whitelist mint slots used).
     */
    function _getAux(address owner) internal view returns (uint64) {
        return _addressData[owner].aux;
    }

    /**
     * Sets the auxillary data for `owner`. (e.g. number of whitelist mint slots used).
     * If there are multiple variables, please pack them into a uint64.
     */
    function _setAux(address owner, uint64 aux) internal {
        _addressData[owner].aux = aux;
    }

    /**
     * Gas spent here starts off proportional to the maximum mint batch size.
     * It gradually moves to O(1) as tokens get transferred around in the collection over time.
     */
    function _ownershipOf(uint256 tokenId) internal view returns (TokenOwnership memory) {
        uint256 curr = tokenId;

        unchecked {
            if (_startTokenId() <= curr && curr < _currentIndex) {
                TokenOwnership memory ownership = _ownerships[curr];
                if (!ownership.burned) {
                    if (ownership.addr != address(0)) {
                        return ownership;
                    }
                    // Invariant:
                    // There will always be an ownership that has an address and is not burned
                    // before an ownership that does not have an address and is not burned.
                    // Hence, curr will not underflow.
                    while (true) {
                        curr--;
                        ownership = _ownerships[curr];
                        if (ownership.addr != address(0)) {
                            return ownership;
                        }
                    }
                }
            }
        }
        revert OwnerQueryForNonexistentToken();
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view override returns (address) {
        return _ownershipOf(tokenId).addr;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        if (!_exists(tokenId)) revert URIQueryForNonexistentToken();

        string memory baseURI = _baseURI();
        return bytes(baseURI).length != 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : '';
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return '';
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public override {
        address owner = ERC721A.ownerOf(tokenId);
        if (to == owner) revert ApprovalToCurrentOwner();

        if (_msgSender() != owner && !isApprovedForAll(owner, _msgSender())) {
            revert ApprovalCallerNotOwnerNorApproved();
        }

        _approve(to, tokenId, owner);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view override returns (address) {
        if (!_exists(tokenId)) revert ApprovalQueryForNonexistentToken();

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        if (operator == _msgSender()) revert ApproveToCaller();

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, '');
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        _transfer(from, to, tokenId);
        if (to.isContract() && !_checkContractOnERC721Received(from, to, tokenId, _data)) {
            revert TransferToNonERC721ReceiverImplementer();
        }
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     */
    function _exists(uint256 tokenId) internal view returns (bool) {
        return _startTokenId() <= tokenId && tokenId < _currentIndex &&
            !_ownerships[tokenId].burned;
    }

    function _safeMint(address to, uint256 quantity) internal {
        _safeMint(to, quantity, '');
    }

    /**
     * @dev Safely mints `quantity` tokens and transfers them to `to`.
     *
     * Requirements:
     *
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called for each safe transfer.
     * - `quantity` must be greater than 0.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(
        address to,
        uint256 quantity,
        bytes memory _data
    ) internal {
        _mint(to, quantity, _data, true);
    }

    /**
     * @dev Mints `quantity` tokens and transfers them to `to`.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `quantity` must be greater than 0.
     *
     * Emits a {Transfer} event.
     */
    function _mint(
        address to,
        uint256 quantity,
        bytes memory _data,
        bool safe
    ) internal {
        uint256 startTokenId = _currentIndex;
        if (to == address(0)) revert MintToZeroAddress();
        if (quantity == 0) revert MintZeroQuantity();

        _beforeTokenTransfers(address(0), to, startTokenId, quantity);

        // Overflows are incredibly unrealistic.
        // balance or numberMinted overflow if current value of either + quantity > 1.8e19 (2**64) - 1
        // updatedIndex overflows if _currentIndex + quantity > 1.2e77 (2**256) - 1
        unchecked {
            _addressData[to].balance += uint64(quantity);
            _addressData[to].numberMinted += uint64(quantity);

            _ownerships[startTokenId].addr = to;
            _ownerships[startTokenId].startTimestamp = uint64(block.timestamp);

            uint256 updatedIndex = startTokenId;
            uint256 end = updatedIndex + quantity;

            if (safe && to.isContract()) {
                do {
                    emit Transfer(address(0), to, updatedIndex);
                    if (!_checkContractOnERC721Received(address(0), to, updatedIndex++, _data)) {
                        revert TransferToNonERC721ReceiverImplementer();
                    }
                } while (updatedIndex != end);
                // Reentrancy protection
                if (_currentIndex != startTokenId) revert();
            } else {
                do {
                    emit Transfer(address(0), to, updatedIndex++);
                } while (updatedIndex != end);
            }
            _currentIndex = updatedIndex;
        }
        _afterTokenTransfers(address(0), to, startTokenId, quantity);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) private {
        TokenOwnership memory prevOwnership = _ownershipOf(tokenId);

        if (prevOwnership.addr != from) revert TransferFromIncorrectOwner();

        bool isApprovedOrOwner = (_msgSender() == from ||
            isApprovedForAll(from, _msgSender()) ||
            getApproved(tokenId) == _msgSender());

        if (!isApprovedOrOwner) revert TransferCallerNotOwnerNorApproved();
        if (to == address(0)) revert TransferToZeroAddress();

        _beforeTokenTransfers(from, to, tokenId, 1);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId, from);

        // Underflow of the sender's balance is impossible because we check for
        // ownership above and the recipient's balance can't realistically overflow.
        // Counter overflow is incredibly unrealistic as tokenId would have to be 2**256.
        unchecked {
            _addressData[from].balance -= 1;
            _addressData[to].balance += 1;

            TokenOwnership storage currSlot = _ownerships[tokenId];
            currSlot.addr = to;
            currSlot.startTimestamp = uint64(block.timestamp);

            // If the ownership slot of tokenId+1 is not explicitly set, that means the transfer initiator owns it.
            // Set the slot of tokenId+1 explicitly in storage to maintain correctness for ownerOf(tokenId+1) calls.
            uint256 nextTokenId = tokenId + 1;
            TokenOwnership storage nextSlot = _ownerships[nextTokenId];
            if (nextSlot.addr == address(0)) {
                // This will suffice for checking _exists(nextTokenId),
                // as a burned slot cannot contain the zero address.
                if (nextTokenId != _currentIndex) {
                    nextSlot.addr = from;
                    nextSlot.startTimestamp = prevOwnership.startTimestamp;
                }
            }
        }

        emit Transfer(from, to, tokenId);
        _afterTokenTransfers(from, to, tokenId, 1);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(
        address to,
        uint256 tokenId,
        address owner
    ) private {
        _tokenApprovals[tokenId] = to;
        emit Approval(owner, to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkContractOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
            return retval == IERC721Receiver(to).onERC721Received.selector;
        } catch (bytes memory reason) {
            if (reason.length == 0) {
                revert TransferToNonERC721ReceiverImplementer();
            } else {
                assembly {
                    revert(add(32, reason), mload(reason))
                }
            }
        }
    }

    /**
     * @dev Hook that is called before a set of serially-ordered token ids are about to be transferred. This includes minting.
     * And also called before burning one token.
     *
     * startTokenId - the first token id to be transferred
     * quantity - the amount to be transferred
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, `from`'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, `tokenId` will be burned by `from`.
     * - `from` and `to` are never both zero.
     */
    function _beforeTokenTransfers(
        address from,
        address to,
        uint256 startTokenId,
        uint256 quantity
    ) internal virtual {}

    /**
     * @dev Hook that is called after a set of serially-ordered token ids have been transferred. This includes
     * minting.
     * And also called after one token has been burned.
     *
     * startTokenId - the first token id to be transferred
     * quantity - the amount to be transferred
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, `from`'s `tokenId` has been
     * transferred to `to`.
     * - When `from` is zero, `tokenId` has been minted for `to`.
     * - When `to` is zero, `tokenId` has been burned by `from`.
     * - `from` and `to` are never both zero.
     */
    function _afterTokenTransfers(
        address from,
        address to,
        uint256 startTokenId,
        uint256 quantity
    ) internal virtual {}
}
// File: contracts/POLYPRIDE.sol

pragma solidity ^0.8.13;

// SPDX-License-Identifier: CC0-1.0
contract POLYPRIDE is ERC721A, ReentrancyGuard, Ownable, Pausable {
    constructor() ERC721A("Polymorphic Pride Flags", "POLYPRIDE") Ownable() {
        // Pause minting until after promoMint()
        _pause();
    }

    uint256 public constant mintPrice = 0.03 ether;
    uint256 public constant maxTokens = 1969;

    // Match to flag #, which starts at 1; array starts at 0
    uint8[1969] currentFlag;

    // Displayed as part of the name & as a trait
    string[12] public flagNames = [
        "Rainbow",
        "Chevron Rainbow",
        "Aromantic",
        "Asexual",
        "Bigender",
        "Bisexual",
        "Genderqueer",
        "Intersex",
        "Lesbian",
        "Non-Binary",
        "Pansexual",
        "Transgender"
    ];

    // All flags start with the opening <svg> tag and the first part of a <path> tag 
    // <svg viewBox="0 0 2000 1200" xmlns="http://www.w3.org/2000/svg">
    //     <path d="
    string public constant flagPrefix = "PHN2ZyB2aWV3Qm94PSIwIDAgMjAwMCAxMjAwIiB4bWxucz0iaHR0cDovL3d3dy53My5vcmcvMjAwMC9zdmciPg0KPHBhdGggZD0iT";

    string[12] public flagData = [
        "TAgMEwyMDAwIDBMMjAwMCAwTDIwMDAgMTIwMEwyMDAwIDEyMDBMMCAxMjAwTDAgMTIwMEwwIDBMMCAwWiIgZmlsbD0iIzc1MDc4NyIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDBMMjAwMCAxMDAwTDIwMDAgMTAwMEwwIDEwMDBMMCAxMDAwTDAgMEwwIDBaIiBmaWxsPSIjMDA0ZGZmIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgMEwyMDAwIDgwMEwyMDAwIDgwMEwwIDgwMEwwIDgwMEwwIDBMMCAwWiIgZmlsbD0iIzAwODAyNiIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDBMMjAwMCA2MDBMMjAwMCA2MDBMMCA2MDBMMCA2MDBMMCAwTDAgMFoiIGZpbGw9IiNmZmVkMDAiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgMEwyMDAwIDBMMjAwMCAwTDIwMDAgNDAwTDIwMDAgNDAwTDAgNDAwTDAgNDAwTDAgMEwwIDBaIiBmaWxsPSIjZmY4YzAwIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgMEwyMDAwIDIwMEwyMDAwIDIwMEwwIDIwMEwwIDIwMEwwIDBMMCAwWiIgZmlsbD0iI2U0MDMwMyIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8L3N2Zz4=",
        "S0wLjY2NjcyMiAwTDE5OTkgMEwxOTk5IDBMMTk5OSAxMjAwTDE5OTkgMTIwMEwtMC42NjY3MjIgMTIwMEwtMC42NjY3MjIgMTIwMEwtMC42NjY3MjIgMEwtMC42NjY3MjIgMFoiIGZpbGw9IiM3NTA3ODciIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTS0wLjY2NjcyMiAwTDE5OTkgMEwxOTk5IDBMMTk5OSAxMDAwTDE5OTkgMTAwMEwtMC42NjY3MjIgMTAwMEwtMC42NjY3MjIgMTAwMEwtMC42NjY3MjIgMEwtMC42NjY3MjIgMFoiIGZpbGw9IiMwMDRkZmYiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTS0wLjY2NjcyMiAwTDE5OTkgMEwxOTk5IDBMMTk5OSA4MDBMMTk5OSA4MDBMLTAuNjY2NzIyIDgwMEwtMC42NjY3MjIgODAwTC0wLjY2NjcyMiAwTC0wLjY2NjcyMiAwWiIgZmlsbD0iIzAwODAyNiIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNLTAuNjY2NzIyIDBMMTk5OSAwTDE5OTkgMEwxOTk5IDYwMEwxOTk5IDYwMEwtMC42NjY3MjIgNjAwTC0wLjY2NjcyMiA2MDBMLTAuNjY2NzIyIDBMLTAuNjY2NzIyIDBaIiBmaWxsPSIjZmZlZDAwIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0tMC42NjY3MjIgMEwxOTk5IDBMMTk5OSAwTDE5OTkgNDAwTDE5OTkgNDAwTC0wLjY2NjcyMiA0MDBMLTAuNjY2NzIyIDQwMEwtMC42NjY3MjIgMEwtMC42NjY3MjIgMFoiIGZpbGw9IiNmZjhjMDAiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTS0wLjY2NjcyMiAwTDE5OTkgMEwxOTk5IDBMMTk5OSAyMDBMMTk5OSAyMDBMLTAuNjY2NzIyIDIwMEwtMC42NjY3MjIgMjAwTC0wLjY2NjcyMiAwTC0wLjY2NjcyMiAwWiIgZmlsbD0iI2U0MDMwMyIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNLTEgMzAyLjM2MkwzMTUuOTQ3IDYwMEwtMSA4OTcuNjM4TC0xIDMwMi4zNjJaIiBmaWxsPSIjZmZmZmZmIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0tMSAzMDIuMzYyTC0xIDE1MS4xODFMNDcyLjU4OCA2MDBMLTEgMTA0OC44MkwtMSA4OTcuNjM4TDMxNS45NDcgNjAwTC0xIDMwMi4zNjJaIiBmaWxsPSIjZmZhZmM4IiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0tMSAxNTEuMTgxTC0xIDBMNjM2LjU2IDYwMEwtMSAxMjAwTC0xIDEwNDguODJMNDcyLjU4OCA2MDBMLTEgMTUxLjE4MVoiIGZpbGw9IiM3NGQ3ZWUiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTS0xIDBMNjM2LjU2IDYwMEwtMSAxMjAwTDE1OS4zMDcgMTIwMEw3OTMuODY4IDYwMEwxNTkuMzA3IDBMLTEgMFoiIGZpbGw9IiM2MTM5MTUiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTE1OS4zMDcgMEwzMTkuMjggMEw5NTMuODQxIDYwMEwzMTkuMjggMTIwMEwxNTkuMzA3IDEyMDBMNzkzLjg2OCA2MDBMMTU5LjMwNyAwWiIgZmlsbD0iIzAwMDAwMCIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8L3N2Zz4=",
        "TAgMEwyMDAwIDBMMjAwMCAxMjAwTDAgMTIwMEwwIDBaIiBmaWxsPSIjMDAwMDAwIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgOTYwTDAgOTYwTDAgMFoiIGZpbGw9IiNhOWE5YTkiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgMEwyMDAwIDBMMjAwMCA3MjBMMCA3MjBMMCAwWiIgZmlsbD0iI2ZmZmZmZiIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDQ4MEwwIDQ4MEwwIDBaIiBmaWxsPSIjYTdkMzc5IiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgMjQwTDAgMjQwTDAgMFoiIGZpbGw9IiMzZGE1NDIiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPC9zdmc+",
        "TAgMEwyMDAwIDBMMjAwMCAxMjAwTDAgMTIwMEwwIDBaIiBmaWxsPSIjZmZmZmZmIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgMzAwTDAgMzAwTDAgMFoiIGZpbGw9IiMwMDAwMDAiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgOTAwTDIwMDAgOTAwTDIwMDAgMTIwMEwwIDEyMDBMMCA5MDBaIiBmaWxsPSIjODAwMDgwIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDMwMEwyMDAwIDMwMEwyMDAwIDYwMEwwIDYwMEwwIDMwMFoiIGZpbGw9IiNhM2EzYTMiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPC9zdmc+",
        "TAgMEwyMDAwIDBMMjAwMCAxNzEuNDI5TDAgMTcxLjQyOUwwIDBaIiBmaWxsPSIjYzQ3OWEyIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDE3MS40MjlMMjAwMCAxNzEuNDI5TDIwMDAgMzQyLjg1N0wwIDM0Mi44NTdMMCAxNzEuNDI5WiIgZmlsbD0iI2VkYTVjZCIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAzNDIuODU3TDIwMDAgMzQyLjg1N0wyMDAwIDg1Ny4xNDNMMCA4NTcuMTQzTDAgMzQyLjg1N1oiIGZpbGw9IiNkNWM3ZTgiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgNTE0LjI4NkwyMDAwIDUxNC4yODZMMjAwMCA2ODUuNzE0TDAgNjg1LjcxNEwwIDUxNC4yODZaIiBmaWxsPSIjZmZmZmZmIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDg1Ny4xNDNMMjAwMCA4NTcuMTQzTDIwMDAgMTAyOC41N0wwIDEwMjguNTdMMCA4NTcuMTQzWiIgZmlsbD0iIzlhYzdlOCIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAxMDI4LjU3TDIwMDAgMTAyOC41N0wyMDAwIDEyMDBMMCAxMjAwTDAgMTAyOC41N1oiIGZpbGw9IiM2ZDgyZDEiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPC9zdmc+",
        "TAgMEwyMDAwIDBMMjAwMCAwTDIwMDAgMTIwMEwyMDAwIDEyMDBMMCAxMjAwTDAgMTIwMEwwIDBMMCAwWiIgZmlsbD0iIzAwMzhhOCIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDBMMjAwMCA3MjBMMjAwMCA3MjBMMCA3MjBMMCA3MjBMMCAwTDAgMFoiIGZpbGw9IiM5YjRmOTYiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgMEwyMDAwIDBMMjAwMCAwTDIwMDAgNDgwTDIwMDAgNDgwTDAgNDgwTDAgNDgwTDAgMEwwIDBaIiBmaWxsPSIjZDYwMjcwIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjwvc3ZnPg==",
        "TAgMEwyMDAwIDBMMjAwMCAxMjAwTDAgMTIwMEwwIDBaIiBmaWxsPSIjNGE4MTIzIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgODAwTDAgODAwTDAgMFoiIGZpbGw9IiNmZmZmZmYiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgMEwyMDAwIDBMMjAwMCA0MDBMMCA0MDBMMCAwWiIgZmlsbD0iI2I1N2VkYyIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8L3N2Zz4=",
        "TAgMEwyMDAwIDBMMjAwMCAxMjAwTDAgMTIwMEwwIDBaIiBmaWxsPSIjZmZkODAwIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik03MDAgNjAwQzcwMCA0MzQuMzE1IDgzNC4zMTUgMzAwIDEwMDAgMzAwQzExNjUuNjkgMzAwIDEzMDAgNDM0LjMxNSAxMzAwIDYwMEMxMzAwIDc2NS42ODUgMTE2NS42OSA5MDAgMTAwMCA5MDBDODM0LjMxNSA5MDAgNzAwIDc2NS42ODUgNzAwIDYwMFoiIGZpbGw9Im5vbmUiIGZpbGwtcnVsZT0iZXZlbm9kZCIgc3Ryb2tlPSIjNzkwMmFhIiBzdHJva2UtbGluZWNhcD0iYnV0dCIgc3Ryb2tlLWxpbmVqb2luPSJtaXRlciIgc3Ryb2tlLXdpZHRoPSIxMDAiLz4NCjwvc3ZnPg==",
        "TAgMEwyMDAwIDBMMjAwMCAwTDIwMDAgMTIwMEwyMDAwIDEyMDBMMCAxMjAwTDAgMTIwMEwwIDBMMCAwWiIgZmlsbD0iI2EzMDI2MiIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDBMMjAwMCA5NjBMMjAwMCA5NjBMMCA5NjBMMCA5NjBMMCAwTDAgMFoiIGZpbGw9IiNkMzYyYTQiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgMEwyMDAwIDBMMjAwMCAwTDIwMDAgNzIwTDIwMDAgNzIwTDAgNzIwTDAgNzIwTDAgMEwwIDBaIiBmaWxsPSIjZmZmZmZmIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgMEwyMDAwIDQ4MEwyMDAwIDQ4MEwwIDQ4MEwwIDQ4MEwwIDBMMCAwWiIgZmlsbD0iI2ZmOWE1NiIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDBMMjAwMCAyNDBMMjAwMCAyNDBMMCAyNDBMMCAyNDBMMCAwTDAgMFoiIGZpbGw9IiNkNTJkMDAiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPC9zdmc+",
        "TAgMEwyMDAwIDBMMjAwMCAxMjAwTDAgMTIwMCIgZmlsbD0iIzJjMmMyYyIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDkwMEwwIDkwMCIgZmlsbD0iIzljNTlkMSIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDYwMEwwIDYwMCIgZmlsbD0iI2ZjZmNmYyIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8cGF0aCBkPSJNMCAwTDIwMDAgMEwyMDAwIDMwMEwwIDMwMCIgZmlsbD0iI2ZjZjQzNCIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8L3N2Zz4=",
        "TAgMEwyMDAwIDBMMjAwMCAxMjAwTDAgMTIwMEwwIDBaIiBmaWxsPSIjMjFiMWZmIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDBMMjAwMCAwTDIwMDAgODAwTDAgODAwTDAgMFoiIGZpbGw9IiNmZmQ4MDAiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgMEwyMDAwIDBMMjAwMCA0MDBMMCA0MDBMMCAwWiIgZmlsbD0iI2ZmMjE4YyIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8L3N2Zz4=",
        "TAgMEwyMDAwIDBMMjAwMCAxMjAwTDAgMTIwMEwwIDBaIiBmaWxsPSIjNWJjZWZhIiBmaWxsLXJ1bGU9ImV2ZW5vZGQiLz4NCjxwYXRoIGQ9Ik0wIDI0MEwyMDAwIDI0MEwyMDAwIDk2MEwwIDk2MEwwIDI0MFoiIGZpbGw9IiNmNWE5YjgiIGZpbGwtcnVsZT0iZXZlbm9kZCIvPg0KPHBhdGggZD0iTTAgNDgwTDIwMDAgNDgwTDIwMDAgNzIwTDAgNzIwTDAgNDgwWiIgZmlsbD0iI2ZmZmZmZiIgZmlsbC1ydWxlPSJldmVub2RkIi8+DQo8L3N2Zz4="
    ];

    function tokenURI(uint256 tokenId) override public view returns (string memory) {
        string memory trait = string(abi.encodePacked('{"trait_type":"Morph","value":"', flagNames[currentFlag[tokenId-1]], '"}'));
        string memory output = string(abi.encodePacked(flagPrefix, flagData[currentFlag[tokenId-1]]));
        
        string memory json = Base64.encode(bytes(string(abi.encodePacked('{"name": "Flag #', toString(tokenId), ' - ', flagNames[currentFlag[tokenId-1]], '", "description": "12-in-1 on-chain LGBTQ+ pride flags. Owner can morph the flag at any time.", "attributes": [', trait ,'], "image": "data:image/svg+xml;base64,', bytes(output), '"}'))));
        output = string(abi.encodePacked('data:application/json;base64,', json));

        return output;
    }

    function weaveFlags(uint8 qty, uint8 flagType) public payable nonReentrant {
        require(qty <= 100, 'EXCEEDS_TX_LIMIT');

        unchecked { 
            require(mintPrice * qty <= msg.value, 'LOW_ETHER');
            require(totalSupply() + qty <= maxTokens, 'MAX_REACHED');

            for(uint256 i = 0; i < qty; ++i) {
                currentFlag[totalSupply()+i] = flagType;
            }

            _safeMint(msg.sender, qty);
        }
    }

    /*
    * Morph the flag into a different design, numbered 0-11:
    *
    *  0 - Simple Rainbow
    *  1 - Chevron Rainbow
    *  2 - Aromantic
    *  3 - Asexual
    *  4 - Bigender
    *  5 - Bisexual
    *  6 - Genderqueer
    *  7 - Intersex
    *  8 - Lesbian
    *  9 - Non-Binary
    * 10 - Pansexual
    * 11 - Transgender
    */
    function morphFlag(uint256 tokenId, uint8 newFlagType) public nonReentrant {
        require(tokenId <= 1969, 'TOKEN_ID_TOO_HIGH');
        require(newFlagType <= 11, 'FLAG_ID_TOO_HIGH');

        unchecked { 
            require(tokenId <= totalSupply(), 'NOT_WOVEN_YET');
            require(ownerOf(tokenId) == msg.sender, 'FLAG_NOT_YOURS');
            currentFlag[tokenId-1] = newFlagType; 
        }
    }

    function _startTokenId() internal pure override returns (uint256) {
        return 1;
    }

    function withdraw() public onlyOwner {
        payable(msg.sender).transfer(address(this).balance);
    }
    
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT license
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /*
    * Mint the first 69 flags to Thorne's wallet. Can only be run once.
    */
    function promoWeave() public onlyOwner {
        unchecked { 
            require(totalSupply() == 0, 'PROMO_RUN'); 

            for(uint256 i = 0; i < 69; ++i) {
                currentFlag[totalSupply()+i] = 11;
            }

            _safeMint(0x8aa986eB2F0D3b5001C9C2093698A4e13d646D5b, 69);
        }
        
        _unpause();
    }

    function pause() public onlyOwner {
        _pause();
    }

    function unpause() public onlyOwner {
        _unpause();
    }
}

/// [MIT License]
/// @title Base64
/// @notice Provides a function for encoding some bytes in base64
/// @author Brecht Devos <brecht@loopring.org>
library Base64 {
    bytes internal constant TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /// @notice Encodes some bytes to the base64 representation
    function encode(bytes memory data) internal pure returns (string memory) {
        uint256 len = data.length;
        if (len == 0) return "";

        // multiply by 4/3 rounded up
        uint256 encodedLen = 4 * ((len + 2) / 3);

        // Add some extra buffer at the end
        bytes memory result = new bytes(encodedLen + 32);

        bytes memory table = TABLE;

        assembly {
            let tablePtr := add(table, 1)
            let resultPtr := add(result, 32)

            for {
                let i := 0
            } lt(i, len) {

            } {
                i := add(i, 3)
                let input := and(mload(add(data, i)), 0xffffff)

                let out := mload(add(tablePtr, and(shr(18, input), 0x3F)))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(shr(12, input), 0x3F))), 0xFF))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(shr(6, input), 0x3F))), 0xFF))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(input, 0x3F))), 0xFF))
                out := shl(224, out)

                mstore(resultPtr, out)

                resultPtr := add(resultPtr, 4)
            }

            switch mod(len, 3)
            case 1 {
                mstore(sub(resultPtr, 2), shl(240, 0x3d3d))
            }
            case 2 {
                mstore(sub(resultPtr, 1), shl(248, 0x3d))
            }

            mstore(result, encodedLen)
        }

        return string(result);
    }
}